

0 to 60 on SPARQL queries in 50 minutes

Ethan Gruber
American Numismatic Society
gruber@numismatics.org
@ewg118



Some URLs

Blog post with list of SPARQL queries in Gist (including slideshow link)

<http://bit.ly/1PGneCf>

Online Coins of the Roman Empire (OCRE)

<http://numismatics.org/ocre/>

Coinage of the Roman Republic Online (CRRO)

<http://numismatics.org/crro/>

Nomisma.org

<http://nomisma.org/>

Google Fusion Tables:

<http://tables.googlelabs.com/>



Denarius
Denier
Δηνάριον



<http://nomisma.org/id/denarius>

Denarius as Triples

```
<http://nomisma.org/id/denarius> skos:prefLabel "Denarius"@en ;  
  rdf:type nmo:Denomination ;  
  skos:exactMatch <http://vocab.getty.edu/aat/300037266> .
```

denarius (nmo:Denomination)

skos:prefLabel	денар (<i>bg</i>), Denari (<i>ca</i>), Denár (<i>cs</i>), Denar (<i>de</i>), Δηνάριον (<i>el</i>), Denarius (<i>en</i>), Denario (<i>es</i>), Denaari (<i>fi</i>), Denier (<i>fr</i>), Dénár (<i>hu</i>), Denario (<i>it</i>), デナリウス (<i>ja</i>), 데나리우스 (<i>ko</i>), Denarius (<i>nl</i>), Denarius (<i>no</i>), Denar (<i>pl</i>), Denário (<i>pt</i>), Denarius (<i>ro</i>), Денарий (<i>ru</i>), Denarius (<i>sv</i>), Денарій (<i>uk</i>), 迪纳厄斯 (<i>zh</i>)
skos:definition	The most common usage is for the main silver coin of the Roman Republic and Empire. The term continued to be used in late imperial accounting and for subsequent coins. Nomisma does not exclude these uses of the term in context where these other senses are intended, though it may develop a more specific vocabulary in the future. (<i>en</i>)
dcterms:isPartOf	http://nomisma.org/id/roman_numismatics
nmo:hasMaterial	http://nomisma.org/id/ar
rdf:type	skos:Concept
skos:exactMatch	http://collection.britishmuseum.org/id/thesauri/currency/denarius
skos:exactMatch	http://dbpedia.org/resource/Denarius
skos:exactMatch	http://vocab.getty.edu/aat/300037266

What is a coin type?



Material: silver
Manufacture: struck
Mint: Emerita
Denomination: Denarius



Authority: Augustus
Legend
Iconography

Defining concepts on nomisma.org



Material: <http://nomisma.org/id/ar>
Manufacture: <http://nomisma.org/id/struck>
Mint: <http://nomisma.org/id/emerita>
Denomination: <http://nomisma.org/id/denarius>



Authority: <http://nomisma.org/id/augustus>
Legend: "Literal"
Icononography: "Literal"

Coin: <http://numismatics.org/collection/1948.19.1029>

Title, collection, accession number

Physical attributes

Image URLs

Findspot

Hoard: <http://numismatics.org/chrr/id/ZAR>

Title, publisher

Findspot

A coin type URI:
[http://numismatics.org/ocre/id/ric.1\(2\).aug.4B](http://numismatics.org/ocre/id/ric.1(2).aug.4B)

<http://collection.britishmuseum.org/id/object/CGR219958>

<http://www.smb.museum/ikmk/object.php?id=18207658>

Displaying records 61 to 68 of 68 total results.

Select from List

Ascending

Sort Results

RRC 467/1a

Date 46 B.C.
Denomination Denarius
Mint Uncertain value
Obverse COS·TERT·DICT·ITER: Head of Ceres right. Border of dots.
Reverse AVGVR PONT·MAX D: Culullus, aspergillum, jug and lituus . Border of dots.



objects: 16; hoards: 52

RRC 521/2

Date 40 B.C.
Denomination Denarius
Mint Uncertain value
Obverse ANT·IMP·III·VIR·R·P·C: Head of M. Antonius right; behind, lituus; around, inscription. Border of dots.
Reverse CN·DOMIT·AHENOBABVS·IMP: Prow right; above, star; around, inscription. Border of dots.



objects: 8; hoard: 1

RRC 540/2

Date 36 B.C.
Denomination Denarius
Mint Uncertain value
Obverse IMP·CAESAR·DIVI·F·III·VIR·ITER·R·P·C: Head of Octavian right, bearded; around, inscription. Border of dots.
Reverse DIVO·IVL COS·ITER·ET·TER·DESIG: Tetrastyle temple; within, figure wearing veil and holding lituus in right hand; on architrave, inscription; within pediment, star; on left, lighted altar; around, inscription. Border of dots.



objects: 21; hoards: 22

RRC 544/14

Date 32 B.C. - 31 B.C.
Denomination Denarius
Mint Uncertain value
Obverse ANT·AVG III·VIR·R·P·C: Ship right, with sceptre tied with fillet on prow . Border of dots.
Reverse LEG II: Aquila between two standards . Border of dots.



objects: 24; hoards: 34

RRC 467/1a

[Examples of this type](#) | [Quantitative Analysis](#)

Typological Description

Date: 46 B.C.
Object Type: [Coin](#)
Manufacture: [Struck](#)
Denomination: [Denarius](#)
Material: [Silver](#)


Authority
Issuer: [Julius Caesar](#)

Geographic
Mint: [Uncertain value](#)

Obverse
Legend: COS·TERT·DICT·ITER
Type
Description: Head of Ceres right. Border of dots.
Deity: [Ceres](#)

Reverse
Legend: AVGVR PONT·MAX D
Type
Description: Culullus, aspergillum, jug and lituus . Border of dots.

Map



Legend ■ Mint ■ Findspot

[View map in fullscreen.](#)

Examples of this type

[RRC 467/1a. 1941.131.310](#)

Collection American Numismatic Society
Axis 1
Weight 3.76



[RRC 467/1a. 1947.2.215](#)

Collection American Numismatic Society
Axis 10
Weight 3.88



[RRC 467/1a. 1896.7.104](#)

Collection American Numismatic Society
Axis 11
Weight 3.39



[RRC 467/1a. 1937.158.277](#)

Collection American Numismatic Society
Axis 2
Weight 3.92



[RRC 467/1a. 1947.2.214](#)

Collection American Numismatic Society
Axis 2
Weight 3.88



[RRC 467/1a. 1970.156.31](#)


Collection American Numismatic Society
Axis 6
Weight 3.95



Quantitative Analysis

Average measurements for this coin type:

Axis: 6.00
Diameter: .00
Weight: 3.61



1. Select Measurement

☐ Axis ☐ Diameter ☒ Weight

2. Select Chart Type

☐ bar ☒ column

3. Compare by Category

☒ Denomination: Denarius ☐ Mint: Caesaraugusta ☐ Region: Lusitania ☐ Manufacture: Struck ☐ Material: Silver ☐ Authority: Augustus
☐ Portrait: Augustus

- Basic intro to SPARQL syntax
- Filtering/sorting
- Dealing with numbers
- OPTIONAL matches

Merging with UNION

- Geographic queries
- Visualization with Google Fusion Tables

http://nomisma.org/sparql

PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>

PREFIX dcterms: <http://purl.org/dc/terms/>

PREFIX skos: <http://www.w3.org/2004/02/skos/core#>

PREFIX owl: <http://www.w3.org/2002/07/owl#>

PREFIX foaf: <http://xmlns.com/foaf/0.1/>

PREFIX ecrm: <http://erlangen-crm.org/current/>

PREFIX geo: <http://www.w3.org/2003/01/geo/wgs84_pos#>

PREFIX nm: <http://nomisma.org/id/>

PREFIX nmo: <http://nomisma.org/ontology#>

PREFIX xsd: <http://www.w3.org/2001/XMLSchema#>

SELECT * WHERE {

 ?s ?p ?o

} LIMIT 100

?s ?p ?o

?s = Subject

?p = Predicate (also, property)

?o = Object

```
SELECT ?label WHERE {  
<http://nomisma.org/id/rome> <http://www.w3.org/2004/02/skos/core#prefLabel> ?label  
}
```

Purpose: get a list of preferred labels for the concept of “Rome.” See <http://nomisma.org/id/rome> for all available properties

https://gist.github.com/ewg118/a5a2de372a7734a090ad#file-rome_labels

<http://nomisma.org/id/rome>

rome (nmo:Mint)

skos:prefLabel Рим (*bg*), Roma (*ca*), Řím (*cs*), Rom (*de*), Ρώμη (*el*), Rome (*en*), Roma (*es*), Rooman (*fi*), Rome (*fr*), Róma (*hu*), Roma (*it*), 古代ローマ (*ja*), 고대 로마 (*ko*), Rome (*nl*), Roma (*no*), Rzym (*pl*), Roma (*pt*), Roma (*ro*), Рим (*ru*), Romerska (*sv*), Рим (*uk*), La Mã cổ đại (*vi*), 古罗马 (*zh*)

skos:definition The mint at the ancient site of Rome in Latium. (*en*)

dcterms:isPartOf http://nomisma.org/id/roman_numismatics

geo:location <http://nomisma.org/id/rome#this>

rdf:type skos:Concept

skos:exactMatch <http://collection.britishmuseum.org/id/place/x25706>

skos:exactMatch http://dbpedia.org/resource/Ancient_Rome

skos:exactMatch <http://geonames.org/3169070>

skos:exactMatch <http://vocab.getty.edu/tn/7594740>

skos:relatedMatch <http://pleiades.stoa.org/places/423025>

#this (geo:SpatialThing)

geo:lat 41.9

geo:long 12.5

Export

[GitHub File](#) [RDF/XML](#) [RDF/TTL](#) [JSON-LD](#) [KML](#)



Unless specified otherwise, content in [Nomisma.org](#) is licensed under a [Creative Commons Attribution 3.0 License](#).

Prefixes: Shortcuts for expressing URIs

PREFIX skos: <http://www.w3.org/2004/02/skos/core#>

PREFIX nm: <http://nomisma.org/id/>

```
SELECT ?label WHERE {  
  nm:rome skos:prefLabel ?label  
}
```

Increasing query complexity

```
SELECT ?label ?type ?field ?lat ?long WHERE {  
  nm:rome skos:prefLabel ?label .  
  nm:rome rdf:type ?type .  
  nm:rome dcterms:isPartOf ?field .  
  nm:rome geo:location ?loc .  
  ?loc geo:lat ?lat .  
  ?loc geo:long ?long  
}
```

Get labels, RDF type (class), field of numismatics, latitude, and longitude
<https://gist.github.com/ewg118/3c55da72510cae200f93>

```
SELECT ?label ?type ?field ?lat ?long WHERE {  
  nm:rome skos:prefLabel ?label ;  
    rdf:type ?type ;  
    dcterms:isPartOf ?field ;  
    geo:location ?loc .  
  ?loc geo:lat ?lat ;  
    geo:long ?long  
}
```

Using semicolons to simplify queries on the same Subject

Sorting and Filtering

```
SELECT ?label WHERE {  
  nm:rome skos:prefLabel ?label  
} ORDER BY ASC(xsd:string(?label))
```

Order by string ?label in ascending (ASC) alphabetical order (DESC for descending)
<https://gist.github.com/ewg118/00be8e5671a795a147ed>

```
SELECT ?label WHERE {  
  nm:rome skos:prefLabel ?label .  
  FILTER langMatches( lang(?label), "en" )  
}
```

Filter only for English labels
<https://gist.github.com/ewg118/8161a9e118f3b8803a97>

Broadening our queries

```
SELECT ?mints WHERE {  
  ?mints a nmo:Mint  
}
```

Get all URIs that are defined as a “mint” in nomisma.org:
<http://nomisma.org/ontology#Mint>

Note: 'a' is equivalent to 'rdf:type'

<https://gist.github.com/ewg118/5e94523486541f3d89e2>

Visualizing Greek Coin Production

```
SELECT ?label ?lat ?long WHERE {  
  ?mints a nmo:Mint ;  
    skos:prefLabel ?label ;  
    dcterms:isPartOf nm:greek_numismatics ;  
    geo:location ?loc .  
  ?loc geo:lat ?lat ;  
    geo:long ?long  
  FILTER langMatches( lang(?label), "en" )  
}
```

Get all mints that are part of Greek numismatics and display the English label, latitude, and longitude

<https://gist.github.com/ewg118/7eb155ed89f04219af0f>

Drops right into Google Fusion Tables!

Let's add complexity: Querying Roman Republican Coins



A Coin Type: **RRC 273/1**
From Michael Crawford's
Roman Republican Coinage (1974)
<http://numismatics.org/crro/id/rrc-273.1>


```
PREFIX crro: <http://numismatics.org/crro/id/>
```

```
SELECT * WHERE {  
  ?objects nmo:hasTypeSeriesItem crro:rrc-273.1 ;  
  a ?type  
}
```

Let's get all associated URIs and their type.

nmo:hasTypeSeriesItem: a URI linking to a reference in a type corpus

<https://gist.github.com/ewg118/e5be0d3b8c3f6c262432>

```
PREFIX crro: <http://numismatics.org/crro/id/>
```

```
SELECT ?objects ?weight WHERE {  
  ?objects nmo:hasTypeSeriesItem crro:rrc-273.1 ;  
    nmo:hasWeight ?weight  
}
```

```
PREFIX crro: <http://numismatics.org/crro/id/>
```

```
SELECT ?objects ?diameter ?weight WHERE {  
  ?objects nmo:hasTypeSeriesItem crro:rrc-273.1 ;  
    nmo:hasWeight ?weight ;  
    nmo:hasDiameter ?diameter  
}
```

Get all of the objects of the coin type RRC 273/1 and display weight/diameter
<https://gist.github.com/ewg118/55f16a2a96521d6e6768>

Optional values

```
SELECT ?objects ?title ?diameter ?weight ?depiction WHERE {  
  ?objects nmo:hasTypeSeriesItem crro:rrc-273.1 ;  
    a nmo:NumismaticObject ;  
    dcterms:title ?title .  
  OPTIONAL { ?objects nmo:hasWeight ?weight }  
  OPTIONAL { ?objects nmo:hasDiameter ?diameter }  
  OPTIONAL { ?objects foaf:depiction ?depiction }  
}
```

Get URIs of RRC 273/1 which are coins. Get the title and optional weight, diameter, and image
<https://gist.github.com/ewg118/9fdcbc32ffa07c7a5f42>

Average Values

```
SELECT (AVG(xsd:decimal(?weight)) AS ?avgWeight) WHERE {  
  ?objects nmo:hasTypeSeriesItem crro:rrc-273.1 ;  
  nmo:hasWeight ?weight  
}
```

What is this doing?

1. Casts ?weight as decimal number
2. Averages these numbers
3. Calls the new variable ?avgWeight

<https://gist.github.com/ewg118/c059341524d187c76185>

Digging Deeper into the Graph

```
SELECT * WHERE {  
  ?types nmo:hasMaterial nm:ar ;  
  dcterms:source nm:rrc  
}
```

All silver coin types from *Roman Republican Coinage*
<https://gist.github.com/ewg118/b7530d6f9156ea2b2f42>

```
SELECT ?objects WHERE {  
  ?types nmo:hasMaterial nm:ar ;  
  dcterms:source nm:rrc .  
  ?objects nmo:hasTypeSeriesItem ?types ;  
  a nmo:NumismaticObject  
} LIMIT 100
```

All silver Republican coins
<https://gist.github.com/ewg118/9500a604b2117698caae>

Extending Queries with UNION

```
SELECT ?objects WHERE {  
  { ?types nmo:hasMaterial nm:ar }  
  UNION { ?types nmo:hasMaterial nm:av }  
  ?types dcterms:source nm:rrc .  
  ?objects nmo:hasTypeSeriesItem ?types ;  
    a nmo:NumismaticObject  
} LIMIT 100
```

Gather all Republican types that are silver and gold and list specimens
<https://gist.github.com/ewg118/8ff575266b7dd1faf114>

Counting

```
SELECT (count(?objects) as ?count) WHERE {  
  ?types nmo:hasMaterial nm:ar ;  
    dterms:source nm:rrc .  
  ?objects nmo:hasTypeSeriesItem ?types ;  
    a nmo:NumismaticObject  
}
```

Above: a count of all silver coins. Below: a count/order by mint

```
SELECT ?label (count(?mint) as ?count) WHERE {  
  ?types nmo:hasMaterial nm:ar ;  
    dterms:source nm:rrc ;  
    nmo:hasMint ?mint .  
  ?mint skos:prefLabel ?label .  
  FILTER langMatches(lang(?label), "en")  
} GROUP BY ?label ORDER by ASC(?label)
```

<https://gist.github.com/ewg118/cb603f187f065acf1535> (above)
<https://gist.github.com/ewg118/c854c94c3ed8fd0af898> (below)

Doing more with geography

```
SELECT DISTINCT ?object ?type ?findspot ?lat ?long ?name WHERE {  
  ?coinType nmo:hasMaterial nm:ar ;  
    dcterms:source nm:ric .  
  { ?object nmo:hasTypeSeriesItem ?coinType ;  
    rdf:type nmo:NumismaticObject ;  
    nmo:hasFindspot ?findspot }  
  UNION { ?object nmo:hasTypeSeriesItem ?coinType ;  
    rdf:type nmo:NumismaticObject ;  
    dcterms:isPartOf ?hoard .  
    ?hoard nmo:hasFindspot ?findspot }  
  UNION { ?contents nmo:hasTypeSeriesItem ?coinType ;  
    a dcmitype:Collection .  
    ?object dcterms:tableOfContents ?contents ;  
    nmo:hasFindspot ?findspot }  
  ?object a ?type .  
  ?findspot geo:lat ?lat .  
  ?findspot geo:long ?long .  
  OPTIONAL { ?findspot foaf:name ?name }  
}
```

Gather a union of all coins with explicit findspots, findspots derived from hoards, and a list of hoards that are connected to silver coin types. Display findspot data.

<https://gist.github.com/ewg118/c1fb8ba5c5609d260205>

Distinct Findspots Only

```
SELECT DISTINCT ?findspot ?lat ?long ?name WHERE {  
  ?coinType nmo:hasMaterial nm:ar ;  
    dcterms:source nm:ric .  
  { ?object nmo:hasTypeSeriesItem ?coinType ;  
    rdf:type nmo:NumismaticObject ;  
    nmo:hasFindspot ?findspot }  
  UNION { ?object nmo:hasTypeSeriesItem ?coinType ;  
    rdf:type nmo:NumismaticObject ;  
    dcterms:isPartOf ?hoard .  
    ?hoard nmo:hasFindspot ?findspot }  
  UNION { ?contents nmo:hasTypeSeriesItem ?coinType ;  
    a dcmitype:Collection .  
    ?object dcterms:tableOfContents ?contents ;  
    nmo:hasFindspot ?findspot }  
  ?object a ?type .  
  ?findspot geo:lat ?lat .  
  ?findspot geo:long ?long .  
  OPTIONAL { ?findspot foaf:name ?name }  
}
```

<https://gist.github.com/ewg118/2dc16c250e77a0996e28>

Filtering By Date

```
SELECT DISTINCT ?object ?type ?findspot ?lat ?long ?name ?closingDate WHERE {  
  ?coinType nmo:hasDenomination nm:denarius ;  
    nmo:hasEndDate ?date .  
  FILTER ( ?date >= "-0100"^^xsd:gYear && ?date <= "-0050"^^xsd:gYear ) .  
  { ?object nmo:hasTypeSeriesItem ?coinType ;  
    rdf:type nmo:NumismaticObject ;  
    nmo:hasFindspot ?findspot }  
  UNION { ?object nmo:hasTypeSeriesItem ?coinType ;  
    rdf:type nmo:NumismaticObject ;  
    dcterms:isPartOf ?hoard .  
    ?hoard nmo:hasFindspot ?findspot }  
  UNION { ?contents nmo:hasTypeSeriesItem ?coinType ;  
    a dcmitype:Collection .  
    ?object dcterms:tableOfContents ?contents ;  
    nmo:hasFindspot ?findspot }  
  ?object a ?type .  
  ?findspot geo:lat ?lat .  
  ?findspot geo:long ?long .  
  OPTIONAL { ?findspot foaf:name ?name }  
  OPTIONAL { ?object nmo:hasClosingDate ?closingDate }  
} ORDER BY ?date
```

<https://gist.github.com/ewg118/0ce62502c1876844ca2a>

Distinct Findspots Only

```
SELECT DISTINCT ?findspot ?lat ?long ?name WHERE {  
  ?coinType nmo:hasMaterial nm:ar ;  
    dcterms:source nm:ric .  
  { ?object nmo:hasTypeSeriesItem ?coinType ;  
    rdf:type nmo:NumismaticObject ;  
    nmo:hasFindspot ?findspot }  
  UNION { ?object nmo:hasTypeSeriesItem ?coinType ;  
    rdf:type nmo:NumismaticObject ;  
    dcterms:isPartOf ?hoard .  
    ?hoard nmo:hasFindspot ?findspot }  
  UNION { ?contents nmo:hasTypeSeriesItem ?coinType ;  
    a dcmitype:Collection .  
    ?object dcterms:tableOfContents ?contents ;  
    nmo:hasFindspot ?findspot }  
  ?object a ?type .  
  ?findspot geo:lat ?lat .  
  ?findspot geo:long ?long .  
  OPTIONAL { ?findspot foaf:name ?name }  
}
```

<https://gist.github.com/ewg118/2dc16c250e77a0996e28>

Filtering By Regular Expression

```
SELECT DISTINCT ?coinTypeLabel ?mintLabel ?lat ?long ?date ?denominationLabel
WHERE {
  ?coinType nmo:hasReverse ?reverse .
  ?reverse dcterms:description ?description FILTER regex(?description, "jupiter", "i") .
  ?coinType skos:prefLabel ?coinTypeLabel FILTER langMatches(lang(?
coinTypeLabel), "en") .
  ?coinType nmo:hasMint ?mint .
  ?mint geo:location ?loc ;
    skos:prefLabel ?mintLabel FILTER langMatches(lang(?mintLabel), "en") .
  ?loc geo:lat ?lat ;
    geo:long ?long .
  ?coinType nmo:hasEndDate ?date FILTER (?date >= "0001"^^xsd:gYear)
  OPTIONAL { ?coinType nmo:hasDenomination ?denomination .
    ?denomination skos:prefLabel ?denominationLabel FILTER
langMatches(lang(?denominationLabel), "en") }
} ORDER BY ASC(?date)
```

Get the geographic distribution of coins mentioning Jupiter on the reverse, after A.D. 1
<https://gist.github.com/ewg118/f5e177c30fff5a282cb2>

Advanced Spatial Query

```
SELECT DISTINCT ?hoard ?label ?lat ?long WHERE {  
  ?loc spatial:nearby (37.974722 23.7225 50 'km') ;  
    geo:lat ?lat ;  
    geo:long ?long .  
  ?hoard nmo:hasFindspot ?loc ;  
    a nmo:Hoard  
    OPTIONAL { ?hoard dcterms:title ?label }  
    OPTIONAL { ?hoard skos:prefLabel ?label }  
}
```

Get hoards found within 50 km of a lat and long (Athens)

<https://gist.github.com/ewg118/0bbf7e3c670f7394665f>